# $Id: PKGBUILD 82849 2013-01-25 01:55:11Z dwallace $
# Maintainer: Vesa Kaihlavirta <vegai@iki.fi>
# Contributor: Kristoffer Fossgård <kfs1@online.no>

pkgname=terminus-font
pkgver=4.38
pkgrel=1
pkgdesc="A superb, monospace bitmap font (for X11 and console)"
arch=('any')
url="http://sourceforge.net/projects/terminus-font/"
license=('GPL2' 'custom:OFL')
depends=('xorg-fonts-encodings' 'xorg-fonts-alias' 'xorg-font-utils' 'fontconfig')
replaces=('xserver-terminus-fonts')
install=terminus-font.install
source=("http://downloads.sourceforge.net/project/${pkgname}/${pkgname}-${pkgver}/${pkgname}-${pkgver}.tar.gz")
md5sums=('a8e792fe6e84c86ed2b6ed3e2a12ba66')

build() { 
  cd "${srcdir}/${pkgname}-${pkgver}"

  sh ./configure --prefix=/usr \
              --x11dir=/usr/share/fonts/local \
              --psfdir=/usr/share/kbd/consolefonts
  make
}
package()
{

  make -C "${pkgname}-${pkgver}" DESTDIR="${pkgdir}" install

  install -Dm644 "${srcdir}/${pkgname}-${pkgver}/OFL.TXT" \
                 "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"

}
